Práctica 2: Planificación del calendario de una liga deportiva

Álvaro Ortiz Villa y Erik Altelarrea Ferré

Programación Matemática (G. Matemáticas, FME). Noviembre 2021

1. Formulación matemática del problema

Para formular el problema definiremos la variable indicadora del partido (i, j) como:

$$x_{i,j,k} = \begin{cases} 1 & \text{si el partido } (i,j) \text{ se juega en la jornada } k \\ 0 & \text{si el partido } (i,j) \text{ no se juega en la jornada } k \end{cases}$$

Definimos, también, los coeficientes $c_{i,j,k}$ como sigue:

$$c_{i,j,k} = \begin{cases} 0 & \text{si el partido } (i,j) \text{ es intra-divisional y } k = 1 \\ 0 & \text{si el partido } (i,j) \text{ es inter-divisional} \\ 2^{k-2} & \text{si el partido } (i,j) \text{ es intra-divisional y } k \geq 2 \end{cases}$$

Mediante los coeficientes anteriores ya podemos definir nuestra **función objetivo** como:

$$z = \sum_{\substack{i,j \in \mathrm{E} \\ i \neq j \\ k \in \mathrm{J}}} c_{i,j,k}\ x_{i,j,k}$$

Así, la formulación del problema es:

$$(PLe) \begin{cases} \text{máx } z = \displaystyle\sum_{\substack{i,j \in \mathrm{E} \\ i \neq j \\ k \in \mathrm{J}}} c_{i,j,k} x_{i,j,k} \\ \qquad \text{s.a.} \\ \qquad\quad \displaystyle\sum_{k \in \mathrm{J}} x_{i,j,k} = r \quad \forall (i,j) \in \text{Partidos intra-divisionales} \\ \qquad\quad \displaystyle\sum_{k \in \mathrm{J}} x_{i,j,k} = s \quad \forall i \in \text{División A}, \forall j \in \text{División B} \\ \qquad\quad \displaystyle\sum_{\substack{j \in \mathrm{E} \\ i \neq j}} x_{i,j,k} = 1 \quad \forall i \in \text{Equipos}, \forall k \in \text{Jornadas} \\ \qquad\quad x_{i,j,k} = x_{j,i,k} \quad \forall i,j \in \text{Equipos}, i \neq j, \forall k \in \text{Jornadas} \end{cases}$$

donde:

- n denota el número de equipos de la liga.
- r denota el número de partidos que juega cada equipo contra cada equipo de su misma división.
- s denota el número de partidos que juega cada equipo contra cada equipo de la otra división.
- $\mathrm{J} = \text{Jornadas} = \{1, \ldots, r(n/2 - 1) + sn/2\}$
- $\text{División A} = \{1, \ldots, n/2\}$
- $\text{División B} = \{n/2 + 1, \ldots, n\}$
- $\mathrm{E} = \text{Equipos} = \{1, \ldots, n\}$
- $\text{Partidos intra-divisionales} = \{(i,j) : i < j, i,j \in \text{División A}\} \cup \{(i,j) : i < j, i,j \in \text{División B}\}$

2. Modelo AMPL desarrollado

Código 1 – Calendario NFL.MOD

```
# PLE calendario NFL

param n >= 0, integer;
param r >= 0, integer;
param s >= 0, integer;
param J = r*((n/2)-1)+s*n/2, integer;

set EQUIPOS:= 1..n;
set JORNADAS:=1..J;

set PARTIDOS_INTRA_DA = {(i, j) in {1..n/2, 1..n/2}: i < j};
set PARTIDOS_INTRA_DB = {(i, j) in {n/2 + 1..n, n/2 + 1..n}: i < j};
set PARTIDOS_INTER = {(i,j) in {1..n/2, n/2 + 1..n}};

var x{(i, j, k) in {EQUIPOS, EQUIPOS, JORNADAS}}, binary;

# Función objetivo
maximize total:
sum {(i, j, k) in {EQUIPOS, EQUIPOS, JORNADAS}: i != j and k >= 2}
if ((i, j) in PARTIDOS_INTRA_DA) or ((i, j) in PARTIDOS_INTRA_DB)
    then x[i,j,k]*(2^(k-2));

# Restricciones

# 1) Partidos a jugar.
# 1.1) Cada equipo de la división A debe jugar r partidos contra cada equipo de su
#división.
subject to res_partidos_DA{(i, j) in PARTIDOS_INTRA_DA}:
sum {k in JORNADAS} (x[i, j, k]) = r;

# 1.2) Cada equipo de la división B debe jugar r partidos contra cada equipo de su
#división.
subject to res_partidos_DB{(i, j) in PARTIDOS_INTRA_DB}:
sum {k in JORNADAS} (x[i, j, k]) = r;

# 1.3) Cada equipo debe jugar s partidos contra cada equipo de la otra división.
subject to res_partidos_inter{(i, j) in PARTIDOS_INTER}:
sum {k in JORNADAS} (x[i, j, k]) = s;

# 2) Cada equipo i juega un único partido cada jornada k.
subject to res_unico_partido{(i, k) in {EQUIPOS, JORNADAS}}:
sum {j in EQUIPOS: j != i} (x[i, j, k]) = 1;

# 3) Para cada jornada k, el partido (i,j) es el mismo que el partido (j,i)
subject to res_orden{(i, j, k) in {EQUIPOS, EQUIPOS, JORNADAS}: j!= i}:
x[i, j, k] = x[j, i, k];
```

Código 2 – Calendario NFL.RUN

```
# Borrar los anterior comandos en AMPL
reset;
# Cargar el modelo
model calendari_NFL.mod;
```

```
# Cargar los datos
data calendari_NFL.dat;
# Indicación a AMPL de que se quiere utilizar como solver: CPLEX
option solver cplex;

#Resolver
solve;

# Mostrar los resultados
printf "\nSOLUCIÓN CASO: n = %d, r = %d, s = %d", n, r, s;
printf "\n——————————————————————";
printf "\n(podrían haber soluciones alternativas con la misma función objectivo) \n";
printf "\nFunción objetivo: %d \n", total;

for {k in JORNADAS} {
  printf "\nPartidos jornada %d", k;
  for {(i,j) in {EQUIPOS, EQUIPOS}: i < j and x[i,j,k] = 1}
  printf "\n(%d,%d)",i,j;
}
printf "\n";
```

3. Soluciones obtenidas

3.1. Conjunto de datos 1

Código 3 – Solución al problema para n = 6, r = 2 y s = 3

```
SOLUCIÓN CASO: n = 6, r = 2, s = 3
——————————————————————
(podrían haber soluciones alternativas con la misma función objectivo)

Función objetivo: 8064

Partidos jornada 1
(1,5)
(2,6)
(3,4)
Partidos jornada 2
(1,6)
(2,4)
(3,5)
Partidos jornada 3
(1,6)
(2,5)
(3,4)
Partidos jornada 4
(1,4)
(2,6)
(3,5)
Partidos jornada 5
(1,5)
```

```
(2,6)
(3,4)
Partidos jornada 6
(1,6)
(2,4)
(3,5)
Partidos jornada 7
(1,5)
(2,4)
(3,6)
Partidos jornada 8
(1,4)
(2,3)
(5,6)
Partidos jornada 9
(1,4)
(2,3)
(5,6)
Partidos jornada 10
(1,2)
(3,6)
(4,5)
Partidos jornada 11
(1,2)
(3,6)
(4,5)
Partidos jornada 12
(1,3)
(2,5)
(4,6)
Partidos jornada 13
(1,3)
(2,5)
(4,6)
```

3.2. Conjunto de datos 2

Código 4 – Solución al problema para n = 8, r = 4 y s = 1

```
SOLUCIÓN CASO: n = 8, r = 4, s = 1
------------------------------------------
(podrían haber soluciones alternativas con la misma función objectivo)

Función objetivo: 131040

Partidos jornada 1
(1,8)
(2,7)
(3,6)
```

```
(4,5)
Partidos jornada 2
(1,7)
(2,5)
(3,8)
(4,6)
Partidos jornada 3
(1,5)
(2,6)
(3,7)
(4,8)
Partidos jornada 4
(1,6)
(2,8)
(3,5)
(4,7)
Partidos jornada 5
(1,2)
(3,4)
(5,6)
(7,8)
Partidos jornada 6
(1,4)
(2,3)
(5,8)
(6,7)
Partidos jornada 7
(1,4)
(2,3)
(5,6)
(7,8)
Partidos jornada 8
(1,3)
(2,4)
(5,7)
(6,8)
Partidos jornada 9
(1,2)
(3,4)
(5,7)
(6,8)
Partidos jornada 10
(1,3)
(2,4)
(5,8)
(6,7)
Partidos jornada 11
(1,4)
```

```
(2,3)
(5,7)
(6,8)
Partidos jornada 12
(1,3)
(2,4)
(5,8)
(6,7)
Partidos jornada 13
(1,3)
(2,4)
(5,7)
(6,8)
Partidos jornada 14
(1,2)
(3,4)
(5,6)
(7,8)
Partidos jornada 15
(1,4)
(2,3)
(5,6)
(7,8)
Partidos jornada 16
(1,2)
(3,4)
(5,8)
(6,7)
```

4. Algunas consideraciones

La función objetivo del modelo AMPL ha sido adaptada según la contribución de los coeficientes de los costes $c_{i,j,k}$. Dado que $c_{i,j,k}$ únicamente contribuye cuando el partido entre (i, j), es intra-divisional y no estamos en la primera jornada, hemos adaptado la función objetivo con tal de que no aparezcan los coeficientes nulos de c.

En el modelo se han definido tres conjuntos de partidos (i, j): dos para los intra-divisionales (Partidos intra-divisionales A= $\{(i, j) : i < j,\ i, j \in \text{División A}\}$ y Partidos intra-divisionales B = $\{(i, j) : i < j,\ i, j \in \text{División B}\}$) y uno para los partidos inter-divisionales (Partidos inter-divisionales = $\{(i, j) : i \in \text{División A},\ j \in \text{División B}\}$). En todo momento guardamos $i < j$ para conservar los partidos en orden lexicográfico.

En cuanto al significado de cada una de las restricciones del (PLe),

- $\sum_{k \in \mathrm{J}} x_{i,j,k} = r$ $\forall (i, j) \in$ Partidos intra-divisionales –o equivalentemente $\forall i, j \in$ División m, con $i \neq j$ y $m =$ A, B– impone que cada equipo juege r partidos contra cada equipo de su división. En el modelo AMPL se impone la restricción para los partidos intra-divisionales de ambas divisiones.

- $\sum_{k \in \mathrm{J}} x_{i,j,k} = s \ \forall i \in$ División A, $\forall j \in$ División B –o lo que es equivalente $\forall (i,j) \in$ Partidos inter-divisionales– impone que cada equipo juege un total de s partidos contra cada equipo de la otra división. En el modelo AMPL se ha utilizado el conjunto de partidos inter-divisionales.

- $\sum_{\substack{j \in \mathrm{E} \\ i \neq j}} x_{i,j,k} = 1 \ \forall i \in$ Equipos, $\forall k \in$ Jornadas impone que cada equipo juegue exactamente un partido por jornada.

- $x_{i,j,k} = x_{j,i,k} \ \forall i,j \in$ Equipos, $i \neq j$, $\forall k \in$ Jornadas impone que en la jornada k, suponiendo que se juegue el partido (i,j), este sea considerado a la vez como (j,i). En general, el valor de la variable indicadora es el mismo para el par (i,j) que para el par (j,i) para cada jornada k.